JOURNAL OF RESEARCH of the National Bureau of Standards
Vol. 86, No. 4, July-August 1981

A Method of Calculating Tunneling Corrections For Eckart Potential Barriers

R. L. Brown*

National Bureau of Standards, Washington, DC 20234

February 18, 1981

A method is presented for the direct calculation of tunneling corrections for unsymmetrical Eckart type potential barriers. It is based on a modified 6-point Gaussian quadrature formula. Accuracy is better than 1 percent over a wide range of tunneling parameter values.

Key words: chemical rate constants; Eckart potential; potential barriers; quantum mechanical tunneling calculations; tunneling corrections.

1. Introduction

The Eckart potential function [1][1] is often used to estimate quantum mechanical tunneling corrections to theoretically determined chemical rate constants. The correction factor Γ^* is defined as the ratio of the quantum mechanical to the classical mechanical barrier crossing rate. It can be expressed [2] as in integral over the energy E,

$$\Gamma^* = \exp(V_1/kT) \int_{E_o}^{\infty} K \exp(-E/kT) dE/kT \qquad (1)$$

where V_1 is the height of the potential barrier, and K is the transmission probability for tunneling. K depends on E and three other parameters which are determined by the shape of the barrier and an effective mass for the system. Johnston and Heicklen [3] have evaluated this integral numerically for a number of parameter values. For certain applications their results are inconvenient to use because interpolation is required to get values not tabulated. In view of this, I have devised a simple method which can be used to calculate Γ^* directly, for any set of parameter values within the ranges chosen by Johnston and Heicklen. The method is presented in the form of a small FORTRAN subroutine called TUNL. In the next section, the details of the method are discussed. Following this, the results of a series of comparisons with an accurate calculation are presented. Finally, the subroutine is listed in the Appendix.

*Center for Thermodynamics and Molecular Science, National Measurement Laboratory.

2. Derivation of the method

Eckart's potential has the form

$$\begin{aligned} V &= -y[A - B/(1-y)]/(1-y) \\ y &= -\exp(2\pi x/L) \\ A &= V_1 - V_2 \\ B &= (V_1^{1/2} + V_2^{1/2})^2 \\ L &= 2\pi(-2/F^*)^{1/2}(V_1^{-1/2} + V_2^{-1/2})^{-1} \end{aligned}$$

The potential has the limiting value of zero when $x \to -\infty$, goes through a single maximum of height V_1 as x increases, and has a limiting value of $V_1 - V_2$ as $x \to +\infty$. F^* is the second derivative of V at its maximum. The lower bound E_0 in the integral (1) is equal to zero when $V_1 \leq V_2$, and to $V_1 - V_2$ when $V_1 > V_2$. The three parameters used by Johnston and Heicklen are α_1, α_2, and u^*.

$$\begin{aligned} u^* &= h\nu^*/kT \\ \alpha_i &= 2\pi V_i/h\nu^*, \; i = 1,2 \\ \nu^* &= (1/2\pi)(-F^*/m)^{1/2} \end{aligned} \qquad (2)$$

where m is an effective mass for tunneling (see ref. 2, p. 53). The integral (1) can be written in a symmetrical form by introducing a new variable, $\epsilon = (E - V_1)/kT$. It becomes

$$\Gamma^* = \int_{\epsilon_o}^{\infty} K e^{-\epsilon} d\epsilon \qquad (3)$$

where $\epsilon_o = -v_1 = -V_1/kT$ when $V_1 \leq V_2$, and $\epsilon_o = -v_2 = -V_2/kT$ when $V_1 > V_2$. In terms of the parameters (2),

the transmission probability K, derived by Eckart, has the form

$$K = [\cosh 2\pi(a_1 + a_2) - \cosh 2\pi(a_1 - a_2)] \times [\cosh 2\pi(a_1 + a_2) + D]^{-1}$$

$$\begin{aligned} 2\pi a_i &= \pi[(\epsilon + v_i)/c]^{1/2},\ i = 1,2 \\ c &= (1/8)\pi u^*(\alpha_1^{-1/2} + \alpha_2^{-1/2})^2 \\ D &= \cosh 2\pi d \text{ if } d \text{ is real} \\ &= \cos 2\pi |d| \text{ if } d \text{ is imaginary} \\ 2\pi d &= (4\alpha_1\alpha_2 - \pi^2)^{1/2} \end{aligned} \tag{4}$$

The method used to evaluate (3) is a modified 6-point Gaussian quadrature formula based on Legendre polynomials [4]. This was used even though the nature of the integral suggests using a formula based on Laguerre polynomials. When the number of evaluations of K is kept small, neither of these methods is satisfactory for the whole range of parameter values used by Johnston and Heicklen, so a variation of the first method was developed.

When ϵ gets large, K approaches unity. The method uses a Gaussian formula for that part of the integral where $K < 1$. The remainder where $K \approx 1$ is evaluated analytically. Thus, if $K(\epsilon) \approx 1$ for $\epsilon > \epsilon_b$, then

$$\int_{\epsilon_b}^{\infty} K(\epsilon) e^{-\epsilon} d\epsilon \approx \int_{\epsilon_b}^{\infty} e^{-\epsilon} d\epsilon = e^{-\epsilon_b}.$$

To evaluate ϵ_b, examine (4) as $\epsilon \to \infty$. One gets $\alpha_i \to \frac{1}{2}(\epsilon/c)^{1/2}$ and

$$K \to 1 - (1 + D)(\tfrac{1}{2}\exp(2\pi(\epsilon_b/c)^{1/2}) + D)^{-1} = K_b.$$

Setting K_b to some value close to unity and solving this equation for ϵ_b gives

$$\epsilon_b = c\left\{\frac{1}{2\pi}\log\left[\frac{2(1 + D)}{1 - K_b}\right]\right\}^2. \tag{5}$$

It happens that this value is not entirely satisfactory, and subtracting from it the average value of v_1 and v_2 gives better results. Also, in some cases, ϵ_b calculated in this way is very large. There is no point in using this value for ϵ_b as the upper bound of the Gaussian formula if the integrand at this point is negligible because of the exponential factor. Thus ϵ_b was kept below a certain fixed value ϵ_{max}. There resulted two parameters, K_b and ϵ_{max}, which were adjusted to minimize the sum of the squares of the differences between the results of this method and the corresponding tabulated values of Johnston and Heicklen.

3. Test of the method

Extensive testing of the accuracy of the method was performed by comparing it with an accurate 40-point Gaussian formula having the cutoff fixed at ϵ_b corresponding to $K_b = 0.999$ or at $\epsilon_b < 8$. In the ranges $0.5 \le \alpha_1 \le \alpha_2 \le 20$, and $2 \le u^* \le 16$, a group of 10,910 comparisons was made. For this set there were the additional restrictions that when $\alpha_1 \ge 8$ then $u^* \le 12$, or when $\alpha_1 \ge 16$ then $u^* \le 10$. Note that $\Gamma^*(\alpha_1, \alpha_2) = \Gamma^*(\alpha_2, \alpha_1)$. A second set of 4,920 comparisons was made in the ranges $0.5 \le \alpha_1 \le \alpha_2 \le 20$, and $0.05 \le u^* \le 1.5$. The results of these tests are given in table 1 in the form of histograms. These show the number of values which differ from the accurate values by a given percentage range. It can be seen from these results that very few values are in error by as much as 5 percent. Such accuracy should be quite adequate for most rate constant calculations.

TABLE I. *Tests of the accuracy of TUNL*

Variation from accurate values	Number of differences in percentage ranges	
Percent difference	Set I [a, b]	Set II [a, c]
−5.5%, −4.5%	3	1
−4.5, −3.5	4	5
−3.5, −2.5	6	19
−2.5, −1.5	30	26
−1.5, −0.5	3246	140
−0.5, 0.5	6811	4475
0.5, 1.5	343	227
1.5, 2.5	217	27
2.5, 3.5	170	
3.5, 4.5	78	
4.5, 5.5	2	
Standard deviations	0.77%	0.42%

[a] For both Sets I and II, $\alpha_2 \ge \alpha_1$. The values used were 0.5, 1.0, 1.5, 2.0, . . . 40.0.

[b] For Set I, $u^* = 2, 3, 4, \ldots 16$. Also if $\alpha_1 \ge 8$ then $u^* \le 12$ and if $\alpha_1 \ge 16$ then $u^* \le 10$.

[c] For Set II, $u^* = 0.05, 0.1, 0.2, 0.5, 1.0, 1.5$.

4. Appendix. Listing of TUNL

The parameters for this program are ALPH1 = α_1, ALPH2 = α_2, $U = u^*$, and $G = \Gamma^*$. It will calculate Γ^* accurately in the parameter ranges $0.5 \le \alpha_1, \alpha_2 \le 20$, and $0 < u^* \le 16$ with the additional restrictions that when α_1 and $\alpha_2 \ge 8$ then $u^* \le 12$, and when α_1 and $\alpha_2 \ge 16$ then $u^* \le 10$.

```
      SUBROUTINE TUNL(ALPH1,ALPH2,U,G)
      DIMENSION X(6),W(6)
      DATA X/-.9324695,-.6612094,-.2386192,.2386192,.6612094,.9324695/
      DATA W/.1713245,.3607616,.4679139,.4679139,.3607616,.1713245/
      DATA PI,PI2,PISQ/3.1415926,6.2831853,9.8696044/
      COSH(Z)=.5*(EXP(Z)+EXP(-Z))
      UPI2=U/PI2
      C=.125*PI*U*(1./SQRT(ALPH1)+1./SQRT(ALPH2))**2
      V1=UPI2*ALPH1
      V2=UPI2*ALPH2
      D=4.*ALPH1*ALPH2-PISQ
      IF(D.LT.0) GOTO 10
      DF=COSH(SQRT(D))
      GOTO 11
   10 DF=COS(SQRT(-D))
   11 IF(V2.GE.V1) EZ=-V1
      IF(V1.GT.V2) EZ=-V2
      EB=AMIN1(C*(ALOG(2.*(1.+DF)/.014)/PI2)**2-.5*(V1+V2),3.2)
      EM=.5*(EB-EZ)
      EP=.5*(EB+EZ)
      G=0
      DO 20 N=1,6
      E=EM*X(N)+EP
      A1=PI*SQRT((E+V1)/C)
      A2=PI*SQRT((E+V2)/C)
      FP=COSH(A1+A2)
      FM=COSH(A1-A2)
   20 G=G+W(N)*EXP(-E)*(FP-FM)/(FP+DF)
      G=EM*G+EXP(-EB)
      RETURN
      END
```

5. References

[1] Eckart, Carl. The penetration of a potential barrier by electrons. Phys. Rev. **35**(11): 1303–1309; 1930 June 1.

[2] Johnston, Harold S. *Gas phase reaction rate theory.* New York: Ronald Press; 1966, 362 p.

[3] Johnston, Harold S.; Heicklen, Julian. Tunneling corrections for unsymmetrical Eckart potential energy barriers. J. Phys. Chem. **66**(3): 532–533; 1962 March.

[4] Abramowitz, M.; Stegun, I. A. Handbook of mathematical functions. Nat. Bur. Stand. (U.S.) Appl. Math. Ser. 55; 1964 June. 1045 p.